# Maintainer: Felix Yan <felixonmars@archlinux.org>
# Contributor: Arch Haskell Team <arch-haskell@haskell.org>

_hkgname=config-ini
pkgname=haskell-config-ini
pkgver=0.2.6.0
pkgrel=18
pkgdesc="A library for simple INI-based configuration files."
url="https://github.com/aisamanra/config-ini"
license=("BSD")
arch=('loong64' 'x86_64')
depends=('ghc-libs' 'haskell-unordered-containers' 'haskell-megaparsec')
makedepends=('ghc' 'uusi' 'haskell-ini' 'haskell-hedgehog' 'haskell-doctest' 'haskell-microlens')
source=("https://hackage.haskell.org/packages/archive/${_hkgname}/${pkgver}/${_hkgname}-${pkgver}.tar.gz")
sha512sums=('33e0f5d6a4cca28c39fc4b37ba00cb0a69c9cd37a75700b2b62bc70e7f2b09abe49bc422bcc7a7e5b2fe48ea3db590c0e3b60fd20ce506004199f903bcbc6ffe')

prepare() {
    cd $_hkgname-$pkgver
    gen-setup
}

build() {
    cd $_hkgname-$pkgver

    runhaskell Setup configure -O --enable-shared --enable-executable-dynamic --disable-library-vanilla \
        --prefix=/usr --docdir=/usr/share/doc/$pkgname --datasubdir=$pkgname --enable-tests \
        --dynlibdir=/usr/lib --libsubdir=\$compiler/site-local/\$pkgid --ghc-option=-fllvm \

# https://github.com/aisamanra/config-ini/issues/38
#            -fenable-doctests
    runhaskell Setup build $MAKEFLAGS
    runhaskell Setup register --gen-script
    runhaskell Setup unregister --gen-script
    sed -i -r -e "s|ghc-pkg.*update[^ ]* |&'--force' |" register.sh
    sed -i -r -e "s|ghc-pkg.*unregister[^ ]* |&'--force' |" unregister.sh
}

check() {
    cd $_hkgname-$pkgver
    runhaskell Setup test --show-details=direct
}

package() {
    cd $_hkgname-$pkgver

    install -D -m744 register.sh "$pkgdir"/usr/share/haskell/register/$pkgname.sh
    install -D -m744 unregister.sh "$pkgdir"/usr/share/haskell/unregister/$pkgname.sh
    runhaskell Setup copy --destdir="$pkgdir"
    install -D -m644 LICENSE "$pkgdir"/usr/share/licenses/$pkgname/LICENSE
    rm -f "$pkgdir"/usr/share/doc/$pkgname/LICENSE
}
